
Subject: FAMILY STRUCTURE

Posted by [Sourav_](#) on Mon, 25 Jul 2022 06:26:08 GMT

[View Forum Message](#) <> [Reply to Message](#)

I want to create several types of family structures and I have tried but unable to do so if you can provide me with STATA code it will be very helpful for me. Thank you. I am attaching the definitions of several types of family structure.

File Attachments

1) [Screenshot \(12\).png](#), downloaded 436 times

2) [Screenshot \(13\).png](#), downloaded 414 times

Subject: Re: FAMILY STRUCTURE

Posted by fred.arnold@icf.com on Mon, 25 Jul 2022 14:09:49 GMT

[View Forum Message](#) <> [Reply to Message](#)

The first thing to recognize is that we are talking about households, not families. All of the information needed comes from the household schedule in the household questionnaire. There can be more than one family in a household, and some households will not include any families if all the persons in the household are not related. The distinction between households and families is very important in any research in this area.

Subject: Re: FAMILY STRUCTURE

Posted by [Sourav_](#) on Sat, 30 Jul 2022 09:28:18 GMT

[View Forum Message](#) <> [Reply to Message](#)

But if we create the household structure based on relationship to the head of household then household become family I guess if it is based on kinship structure.

Subject: Re: FAMILY STRUCTURE

Posted by fred.arnold@icf.com on Sat, 30 Jul 2022 16:15:24 GMT

[View Forum Message](#) <> [Reply to Message](#)

The household does not always become a family. The following three response categories for relationship to household head would not be considered part of a family:

Domestic servant

Other not related

Don't know

In some households, none of the household members are related to head of the household.

Subject: Re: FAMILY STRUCTURE
Posted by [Sourav_](#) on Thu, 29 Dec 2022 08:04:29 GMT
[View Forum Message](#) <> [Reply to Message](#)

Yes. By excluding this three category of member (domestic servant, other not related and don't know), can you please help me with code or guide me how to create the types of family which is attached?

File Attachments

- 1) [Screenshot \(12\).png](#), downloaded 393 times
 - 2) [Screenshot \(13\).png](#), downloaded 367 times
-

Subject: Re: FAMILY STRUCTURE
Posted by fred.arnold@icf.com on Mon, 02 Jan 2023 17:34:35 GMT
[View Forum Message](#) <> [Reply to Message](#)

It's best not to refer to persons living in a household as a family, since many members of a family could be living elsewhere. It's also important to include only persons who usually live in the household and not to include visitors.

In the NFHS-5 report, we only break the household down into two groups (nuclear households and non-nuclear households). The definition used is as follows:
"Nuclear households are households comprised of a married couple or a man or a woman living alone or with unmarried children (biological, adopted, or fostered) with or without unrelated individuals." We do not have any code or guidance for creating the groups that you are trying to define.

Another possible concern is how to define the household structure in households in which the household head has more than one wife living in the same household.

Subject: Re: FAMILY STRUCTURE
Posted by [Sourav_](#) on Tue, 03 Jan 2023 04:50:56 GMT
[View Forum Message](#) <> [Reply to Message](#)

Ok sir. Thank you very much for your reply.

Subject: Re: FAMILY STRUCTURE
Posted by [Sourav_](#) on Sat, 14 Jan 2023 11:13:49 GMT
[View Forum Message](#) <> [Reply to Message](#)

Sir, I have a query. Is it possible to get a single father family in NFHS means only father with unmarried child or it is unavailable?

Subject: Re: FAMILY STRUCTURE

Posted by [Bridgette-DHS](#) on Tue, 17 Jan 2023 16:38:08 GMT

[View Forum Message](#) <> [Reply to Message](#)

Following is a response from Senior DHS staff member, Tom Pullum:

You need to construct a typology of household types, of which this specific household structure would be one of the types, using the PR file. I suggest that the dimensions are (1) the sex of the household head, (2) whether a spouse is present, and (3) whether a dependent child (age 0-17 and unmarried) is present. There are $2 \times 2 \times 2 = 8$ combinations. You first construct a binary variable to describe each household member based on relation to head, sex, age, or marital status, then add up the number of people of these types within the household and assign these subtotals to each person in the household. You then construct the typology. Then you can reduce to the household head or do tabulations for just the head, so the units are households rather than individuals. I will paste below the Stata code to do this. The household type you are specifically interested in is #3. I will then paste the distribution. There is a small "other" category, probably because hv105 includes a "transgender" category. You may want to restrict to de jure household members--I didn't do that.

Type 3 represents about $\frac{1}{2}$ of 1 percent of all households. Type 7, with a female head, no spouse, and dependent children, represents about 3% of households

use "...IAPR7DFL.DTA", clear

- * Criteria for household typology
- * household head is male (hv101=1, hv104=1)
- * household head is female (hv101=1, hv104=2)
- * no spouse present (no one in hh with hv101=2)
- * at least one unmarried child present (hv101=3, hv105<=17, hv115=0)

```
gen hhhead_male =0
gen hhhead_female=0
gen spouse=0
gen child=0
```

```
replace hhhead_male =1 if hv101==1 & hv104==1
replace hhhead_female=1 if hv101==1 & hv104==2
replace spouse=1      if hv101==2
replace child=1       if hv101==3 & hv105<=17 & hv115==0
```

```
egen nhhhead_male =total(hhhead_male), by(hv024 hv001 hv002)
egen nhhhead_female=total(hhhead_female), by(hv024 hv001 hv002)
egen nspouse      =total(spouse),      by(hv024 hv001 hv002)
egen nchild       =total(child),       by(hv024 hv001 hv002)
```

```
gen hhtype=.
replace hhtype=1 if nhhhead_male==1 & nspouse>=1 & nchild>0
replace hhtype=2 if nhhhead_male==1 & nspouse>=1 & nchild==0
replace hhtype=3 if nhhhead_male==1 & nspouse==0 & nchild>0
```

```
replace hhtype=4 if nhhhead_male==1 & nspouse==0 & nchild==0
replace hhtype=5 if nhhhead_female==1 & nspouse>=1 & nchild>0
replace hhtype=6 if nhhhead_female==1 & nspouse>=1 & nchild==0
replace hhtype=7 if nhhhead_female==1 & nspouse==0 & nchild>0
replace hhtype=8 if nhhhead_female==1 & nspouse==0 & nchild==0
replace hhtype=9 if hhtype==.
```

```
label define hhtype 1 "Male head with spouse and children" 2 "Male head with spouse, no
children" 3 "Male head, no spouse, and children" 4 "Male head, no spouse, no children" 5 "Female
head with spouse and children" 6 "Female head with spouse, no children" 7 "Female head, no
spouse, and children" 8 "Female head, no spouse, no children" 9 "Other"
label values hhtype hhtype
```

```
tab hhtype if hv101==1 [iweight=hv005/1000000
```

File Attachments

1) [hhtype..jpg](#), downloaded 1193 times

Subject: Re: FAMILY STRUCTURE

Posted by [Sourav_](#) on Wed, 18 Jan 2023 06:52:27 GMT

[View Forum Message](#) <> [Reply to Message](#)

yes I got the idea. But all of hhtypes which you have created may have other members in them also right?

Subject: Re: FAMILY STRUCTURE

Posted by [Bridgette-DHS](#) on Wed, 18 Jan 2023 12:46:19 GMT

[View Forum Message](#) <> [Reply to Message](#)

Following is a response from Senior DHS staff member, Tom Pullum:

Yes, in this typology, other household members may be present or absent. You could include checks for the presence or absence of other people based on the relation to head code, age, and sex.

Subject: Re: FAMILY STRUCTURE

Posted by [Sourav_](#) on Wed, 18 Jan 2023 15:56:08 GMT

[View Forum Message](#) <> [Reply to Message](#)

Ok. Thank you very much for your response.

Subject: Re: FAMILY STRUCTURE

Posted by fred.arnold@icf.com on Wed, 18 Jan 2023 15:56:23 GMT

[View Forum Message](#) <> [Reply to Message](#)

Yes, they may have other household members, so you have to decide whether to include them or exclude them in your analysis.

Subject: Re: FAMILY STRUCTURE

Posted by [Sourav_](#) on Wed, 02 Aug 2023 05:36:02 GMT

[View Forum Message](#) <> [Reply to Message](#)

1. The household structure can be formed from both PR file and HR file or only from PR file?
 2. If I want to see the maternal health of women across several household structures then is it right to form the household structure in any of the PR or HR files and then merge that file to IR file or should I have to create the household structure directly in IR file?
-

Subject: Re: FAMILY STRUCTURE

Posted by [Bridgette-DHS](#) on Wed, 02 Aug 2023 17:26:48 GMT

[View Forum Message](#) <> [Reply to Message](#)

Following is a response from Senior DHS staff member, Tom Pullum:

You can construct indicators of household structure using either the PR file (with the egen commands in Stata) or with the HR file (looping over the index for household members). If you do it with the HR file, you will have one record per household and you can merge it with the woman in the IR file by matching hv001 hv002 with v001 v002. If you do it with the PR file, the structure codes will be the same for everyone in the same household. You can reduce to one record per household with "keep if hvidx==1". Then merge with the IR file by matching hv001 hv002 with v001 v002. Let us know if you have more questions about constructing indicators of household structure. There have been some earlier forum post on this topic.

Subject: Re: FAMILY STRUCTURE

Posted by [Sourav_](#) on Sun, 06 Aug 2023 10:52:43 GMT

[View Forum Message](#) <> [Reply to Message](#)

Can you please show the same household structure for PR and HR file separately with different command as you are saying that in PR with egen command and in HR with loop. Like if we want

to create a household "head without spouse but with unmarried children and with other relations of whom only one is having spouse" how will we do for both the files?

What are the pros and cons of this two types of files for making household structure variable and PR file is prefer over HR file for making household structure variable for what?

Subject: Re: FAMILY STRUCTURE

Posted by [Bridgette-DHS](#) on Mon, 14 Aug 2023 15:50:43 GMT

[View Forum Message](#) <> [Reply to Message](#)

Following is a response from Senior DHS staff member, Tom Pullum:

We apologize for the delay in this reply, due to travel. I don't have time to give two solutions to the same problem so will just show how you can develop measures of household structure using egen and the PR file. You will end up with one line per household. Measures of structure should be limited to household members who are de jure. If the household head is not de jure (yes, this can happen) then the household drops out. I work out a slightly different example, but you should be able to generalize it.

The first step is to construct binary variables for the relation to head codes (hv101), with extensions (if needed) to include marital status or other characteristics. Then you add up those binary variables within the household, using egen total, to see whether certain types of potential household members are, or are not, present. Those totals will be added to every case in the household. Then reduce to just the household head to get one record per household. Let us know if you have further questions.

- * Identify households with this structure:
- * Spouse of head not present; children of head present; all children of head are unmarried
- * Household structure is defined entirely in terms of de jure household members (hv102=1)

* head: hv101=1; spouse, hv101=2, child: hv101=3

* unmarried: hv115=0

use "...IAPR7EFL.DTA", clear

keep if hv102==1

gen spouse=0

replace spouse=1 if hv101==2

gen child=0

replace child=1 if hv101==3

gen unmarried_child=0

replace unmarried_child=1 if hv101==3 & hv115==0

```
egen nspouse      =total(spouse),      by(hv024 hv001 hv002)
egen nchild       =total(child),       by(hv024 hv001 hv002)
egen nunmarried_child=total(unmarried_child), by(hv024 hv001 hv002)
```

```
* reduce to one line per household
keep if hv101==1
drop spouse child unmarried_child
gen sex_of_head=hv104
gen age_of_head=hv105
gen spouse_absent=0
replace spouse_absent=1 if nspouse==0
```

```
rename nchild nchildren
rename nunmarried_child nunmarried_children
gen nmarried_children=nchildren-nunmarried_children
```

```
tab sex_of_head spouse_absent if nchildren>0 & nmarried_children==0
```

* The households with the desired structure will be those with
* spouse_absent=1, nchildren>0, and nmarried_children=0.

Subject: Re: FAMILY STRUCTURE
Posted by [Sourav_](#) on Sat, 19 Aug 2023 06:48:56 GMT
[View Forum Message](#) <> [Reply to Message](#)

Can you please tell me how can I find exactly who are couple in HR or PR file of DHS? Like there are two sons and one daughter-in-law in household and how can I found who are couple? Like three of them can be married and any one of their spouse may not be the usual residents of that household or may it be like there are two married son and one married daughter-in-law in household but actually the daughter-in-law is spouse of other married son who is not usual residents of the household and as in DHS those members are present in household only asks if they are usual residents or not. Like I am saying, if a household have actually three married son and one married daughter-in-law but only two married son is living with other daughter-in-law whose spouse is not usual residents of the household. Can you help me out how to think of proceed for this case? How can I find who are exactly couple except head and spouse in the household?

Subject: Re: FAMILY STRUCTURE
Posted by [Bridgette-DHS](#) on Mon, 21 Aug 2023 16:20:49 GMT
[View Forum Message](#) <> [Reply to Message](#)

Following is a response from Senior DHS staff member, Tom Pullum:

The relation to head code (hv101) in the household (PR) file cannot identify any couples other than the head+spouse couple.

The Guide to DHS Statistics (<https://www.dhsprogram.com/Data/Guide-to-DHS-Statistics/index.cfm>) includes a description of how the couples file (the CR file) is constructed. The woman (in the IR file) and the man (in the MR file) must specify each other. If you go to the CR file, you will find all the couples that can be constructed using this requirement. It should include virtually all of the head+spouse couples

In the India surveys, only one-sixth of households were selected for the men's survey. Outside of that one-sixth of the total sample, no couples other than head+spouse can be identified.

Subject: Re: FAMILY STRUCTURE

Posted by [Sourav_](#) on Mon, 21 Aug 2023 16:28:30 GMT

[View Forum Message](#) <> [Reply to Message](#)

Ok. Couples can not be identified using the line numbers of the husbands of IR file merging with PR file?
